

# NAPREDNI KURS

## Alaska Xbase++

### Resource ICO, BMP, PNG, GIF, JPG for Alska Xbase++

#### Upotreba u Alaska Xbase++ aplikaciji

Uz ovaj tekst ide kompletan projekat EXE i DLL programa u fajlu RESOURCE.ZIP koji se nalazi na adresi: [www.cobasystems.com/download/resource.zip](http://www.cobasystems.com/download/resource.zip)

*COBA Systems*

## Resursi za Alaska Xbase++

Ovde se obrađuju samo Resursi u koje spadaju ikone i bitmape: BMP, PNG, GIF i JPG

Primeri su dati i provereni za verzije programa:

### **Alaska Xbase++**

Release: 1.90.355

Service Level #1 for Xbase++ Version 1.90.355  
for Windows 32Bit and 64Bit Operating Systems  
International Edition, Date: May 20th 2009

### **eXpress++**

Version 1.9 build 259.

This version supports or Xbase++ 1.9 (build 331/355)  
and Xbase++ 2.0 (build 472).

Autor

Slobodan Stanojević Coba, dipl.ing.

[www.cobasystems.com](http://www.cobasystems.com)

# **IKONE**

## **SECTION: ICON**

### **(FileName.ICO)**

## 1. 1. Upotreba ikone koja se nalazi kao fajl **TEST.ICO** na disku računara

```
* Xbase++
oICO := XbpIcon():new():create()
oICO:LoadFile("TEST.ICO")

* PushButton
oButon := XbpPushButton():new()
oButon:caption := oICO
oButon:create( oDlg, , {10,20}, {70,70} )

* Icon
oIcon:= XbpStatic():new(oDlg,oDlg, {30,100}, {32,32} )
oIcon:type := XBPSTATIC_TYPE_ICON
oIcon:caption := oICO
oIcon:autoSize := .T.
oIcon:create()

// Ikona je transparentna na PushButton i na Icon
```

```
* eXpress++
oICO := DC_GetBitmap( "TEST.ICO",,,,, .T. )

* PushButton
@ 1,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3

* Icon
@ 1,1 DCSTATIC TYPE XBPSTATIC_TYPE_ICON SIZE 32,32 PIXEL CAPTION oICO

// Ikona je transparentna na PushButton i na Icon
```

## 1.2.1. Upotreba ikone koja se nalazi kao resurs broj 1 u EXE fajlu aplikacije

```
* Xbase++
oICO := XbpIcon():new():create()
oICO:Load(NIL,1)

* PushButton
oButon := XbpPushButton():new()
oButon:caption := oICO
oButon:create( oDlg, , {10,20}, {70,70} )

* Icon
oIcon:= XbpStatic():new(oDlg,oDlg, {30,100}, {32,32} )
oIcon:type := XBPSTATIC_TYPE_ICON
oIcon:caption := oICO
oIcon:autoSize := .T.
oIcon:create()

// Ikona je transparentna na PushButton i na Icon
```

```
* eXpress++
oICO := DC_GetBitmap( 1 ,,,, , .T. )

* PushButton
@ 1,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3

* Icon
@ 1,1 DCSTATIC TYPE XBPSTATIC_TYPE_ICON SIZE 32,32 PIXEL CAPTION oICO

// Ikona je transparentna na PushButton i na Icon
```

1.2.2. Upotreba ikone koja se nalazi kao resurs **TEST\_ICO** i broj **1** u EXE fajlu a u istom EXE fajlu poziva se i koristi se i **RESURS.CH** fajl koji sadrži: **#define TEST\_ICO 1**

```
* Xbase++
* Icon
oIcon:= XbpStatic():new(oDlg,oDlg, {30,100}, {32,32} )
oIcon:type := XBPSTATIC_TYPE_ICON
oIcon:caption := 1
// oIcon:caption := TEST_ICO
oIcon:autoSize := .T.
oIcon:create()

// upotrebljava se ili 1 ili TEST_ICO
// Ne može se koristiti za PushButton
// Ikona je transparentna na Icon
```

```
* eXpress++
* Icon
@ 1,1 DCSTATIC TYPE XBPSTATIC_TYPE_ICON SIZE 32,32 PIXEL CAPTION 1

// upotrebljava se ili 1 ili TEST_ICO
// Ne može se koristiti za PushButton
// Ikona je transparentna na Icon
```

### 1.3.1. Upotreba ikone koja se nalazi kao resurs broj 1 u RESURS.DLL fajlu

```
* Xbase++
oICO := XbpIcon():new():create()
oICO:Load("RESURS.DLL",1)

* PushButton
oButon := XbpPushButton():new()
oButon:caption := oICO
oButon:create( oDlg, , {10,20}, {70,70} )

* Icon
oIcon:= XbpStatic():new(oDlg,oDlg, {30,100}, {32,32} )
oIcon:type := XBPSTATIC_TYPE_ICON
oIcon:caption := oICO
oIcon:autoSize := .T.
oIcon:create()
// Ikona je transparentna na PushButton i na Icon
```

```
* eXpress++
oICO := DC_GetBitmap( 1 ,, "ICO" , "RESURS.DLL" )

* PushButton
@ 1,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3

* Icon
@ 1,1 DCSTATIC TYPE XBPSTATIC_TYPE_ICON SIZE 32,32 PIXEL CAPTION oICO

// Ikona je transparentna na PushButton i na Icon
```

### 1.3.2. Upotreba ikone koja se nalazi kao resurs broj 1 u RESURS.DLL fajlu

```
* Xbase++
* Icon
oIcon:= XbpStatic():new(oDlg,oDlg, {30,100}, {32,32} )
oIcon:type := XBPSTATIC_TYPE_ICON
oIcon:SetCaption := (1,"RESURS.DLL")
oIcon:autoSize := .T.
oIcon:create()

// Ne može se koristiti za PushButton
// Ikona je transparentna na Icon
```

# **BITMAPE**

## **SECTION: BITMAP**

### **(FileName.BMP)**



## 2. 1. Upotreba bitmape koja se nalazi kao fajl **TEST.BMP** na disku računara

```
* Xbase++
oBMP := XbpBitmap():new():create()
oBMP:LoadFile("TEST.BMP")
oBMP:transparentClr := oBMP:getDefaultBgColor()

* PushButton
oButon := XbpPushButton():new()
oButon:caption := oBMP
oButon:create( oDlg, , {10,20}, {70,70} )

* Picture
oBitmap:= XbpStatic():new(oDlg,oDlg, {30,100}, {50,50} )
oBitmap:type := XBPSTATIC_TYPE_BITMAP
oBitmap:caption := oBMP
oBitmap:autoSize := .T.
oBitmap:create()

// Sa oBMP:transparentClr bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna
```

```
* eXpress++
aColor := {199,199,199} // BMP transparent RGB color gray
DC_BitmapTransparentColor(aColor)

oBMP := DC_GetBitmap( "TEST.BMP" )

* PushButton
@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3

* Picture
@ 1,1 DCSTATIC TYPE XBPSTATIC_TYPE_BITMAP SIZE 50,50 PIXEL CAPTION oBMP

// Sa DC_BitmapTransparentColor(aColor) bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna
```

2. 2. 1. Upotreba bitmape koja se nalazi kao **resurs broj 101** u **EXE** fajlu aplikacije

```
* Xbase++
oBMP := XbpBitmap():new():create()
oBMP:LoadFile(NIL,101)
oBMP:transparentClr := oBMP:getDefaultBgColor()

* PushButton
oButon := XbpPushButton():new()
oButon:caption := oBMP
oButon:create( oDlg, , {10,20}, {70,70} )

* Picture
oBitmap:= XbpStatic():new(oDlg,oDlg, {30,100}, {50,50} )
oBitmap:type := XBPSTATIC_TYPE_BITMAP
oBitmap:caption := oBMP
oBitmap:autoSize := .T.
oBitmap:create()

// Sa oBMP:transparentClr bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna
```

```
* eXpress++
aColor := {199,199,199} // BMP transparent RGB color gray
DC_BitmapTransparentColor(aColor)

oBMP := DC_GetBitmap( 101 )

* PushButton
@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3

* Picture
@ 1,1 DCSTATIC TYPE XBPSTATIC_TYPE_BITMAP SIZE 50,50 PIXEL CAPTION oBMP

// Sa DC_BitmapTransparentColor(aColor) bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna
```

2. 2. 2. Upotreba bitmape koja se nalazi kao **resurs TEST\_BMP broj 101** u **EXE** fajlu aplikacije a u istom EXE fajlu poziva se i koristi se i **RESURS.CH** fajl koji sadrži:

**#define TEST\_BMP 101**

```
* Xbase++
oBMP:transparentClr := oBMP:getDefaultBgColor()

* PushButton
oButon := XbpPushButton():new()
oButon:caption := 101
oButon:create( oDlg, , {10,20}, {70,70} )

* Picture
oBitmap:= XbpStatic():new(oDlg,oDlg, {30,100}, {50,50} )
oBitmap:type := XBPSTATIC_TYPE_BITMAP
oBitmap:caption := 101
oBitmap:autoSize := .T.
oBitmap:create()

// upotrebljava se ili 101 ili TEST_BMP
// Sa oBMP:transparentClr bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna
```

```
* eXpress++
aColor := {199,199,199} // BMP transparent RGB color gray
DC_BitmapTransparentColor(aColor)

* PushButton
@ 1,1 DCPUSHBUTTON CAPTION 101 SIZE 8,3

* Picture
@ 1,1 DCSTATIC TYPE XBPSTATIC_TYPE_BITMAP SIZE 50,50 PIXEL CAPTION 101

// upotrebljava se ili 101 ili TEST_BMP
// Sa DC_BitmapTransparentColor(aColor) bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna
```

### 2.3. Upotreba bitmape koja se nalazi kao **resurs broj 101** u **RESURS.DLL** fajlu

```
* Xbase++
oBMP := XbpBitmap():new():create()
oBMP:LoadFile("RESURS.DLL",101)
oBMP:transparentClr := oBMP:getDefaultBgColor()

* PushButton
oButon := XbpPushButton():new()
oButon:caption := oBMP
oButon:create( oDlg, , {10,20}, {70,70} )

* Picture
oBitmap:= XbpStatic():new(oDlg,oDlg, {30,100}, {50,50} )
oBitmap:type := XBPSTATIC_TYPE_BITMAP
oBitmap:caption := oBMP
oBitmap:autoSize := .T.
oBitmap:create()

// Sa oBMP:transparentClr bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna
```

```
* eXpress++
aColor := {199,199,199} // BMP transparent RGB color gray
DC_BitmapTransparentColor(aColor)

oBMP := DC_GetBitmap( 101 ,, "BMP" , "RESURS.DLL" )

* PushButton
@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3

* Picture
@ 1,1 DCSTATIC TYPE XBPSTATIC_TYPE_BITMAP SIZE 50,50 PIXEL CAPTION oBMP

// Sa DC_BitmapTransparentColor(aColor) bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna
```

**BITMAPE**  
**SECTION: USERDEF**  
**(FileName.PNG, FileName.GIF,**  
**FileName.JPG)**

### 3. 1. Upotreba bitmape koja se nalazi kao fajl **TEST.PNG** na disku računara

```
* Xbase++
oBMP := XbpBitmap():new():create()
oBMP:LoadFile("TEST.PNG") // or "TEST.GIF" or "TEST.JPG"
oBMP:transparentClr := oBMP:getDefaultBgColor() // for TEST.JPG

* PushButton
oButon := XbpPushButton():new()
oButon:caption := oBMP
oButon:create( oDlg, , {10,20}, {70,70} )

* Picture
oBitmap:= XbpStatic():new(oDlg,oDlg, {30,100}, {50,50} )
oBitmap:type := XBPSTATIC_TYPE_BITMAP
oBitmap:caption := oBMP
oBitmap:autoSize := .T.
oBitmap:create()

// Sa oBMP:transparentClr JPG bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna.
// Bitmape PNG i GIF transparentne su bez komande oBMP:transparentClr
```

```
* eXpress++
aColor := {0,0,0} // PNG transparent RGB color black
DC_BitmapTransparentColor(aColor)

oBMP := DC_GetBitmap( "TEST.PNG" ) // or "TEST.GIF" or "TEST.JPG"

* PushButton
@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3

* Picture
@ 1,1 DCSTATIC TYPE XBPSTATIC_TYPE_BITMAP SIZE 50,50 PIXEL CAPTION oBMP

// Sa DC_BitmapTransparentColor(aColor) JPG bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna.
// Bitmape PNG i GIF transparentne su bez komande
// DC_BitmapTransparentColor(aColor)
```

3. 2 Upotreba bitmape koja se nalazi kao **resurs broj 1001** u **EXE** fajlu aplikacije a koja je definisana u ARC fajlu resursa kao:

**USERDEF PNG**

**1001 = FILE "TEST.PNG"**

```
* Xbase++
oBMP := XbpBitmap():new():create()
cBuffer := LoadResource( 1001, "PNG" ) // "BMP", "PNG", "GIF", "JPG"
oBMP:SetBuffer( cBuffer, XBPBMP_FORMAT_PNG ) // _BMP, _PNG, _GIF, _JPG
oBMP:transparentClr := oBMP:getDefaultBgColor()

* PushButton
oButon := XbpPushButton():new()
oButon:caption := oBMP
oButon:create( oDlg, , {10,20}, {70,70} )

* Picture
oBitmap:= XbpStatic():new(oDlg,oDlg, {30,100}, {50,50} )
oBitmap:type := XBPSTATIC_TYPE_BITMAP
oBitmap:caption := oBMP
oBitmap:autoSize := .T.
oBitmap:create()

// Sa oBMP:transparentClr bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna
```

```
* eXpress++
aColor := {0,0,0} // PNG transparent RGB color black
DC_BitmapTransparentColor(aColor)

oBMP := DC_GetBitmap( 1001, "PNG" ) // "BMP", "PNG", "GIF", "JPG"

* PushButton
@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3

* Picture
@ 1,1 DCSTATIC TYPE XBPSTATIC_TYPE_BITMAP SIZE 50,50 PIXEL CAPTION oBMP

// Sa DC_BitmapTransparentColor(aColor) bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna
```

3.3 Upotreba bitmape koja se nalazi kao **resurs broj 1001** u **RESURS.DLL** fajlu a koja je definisana u ARC fajlu resursa kao:

**USERDEF PNG**

**1001 = FILE "TEST.PNG"**

```
* Xbase++
oBMP := XbpBitmap():new():create()
cBuffer := LoadResource(1001,"RESURS.DLL","PNG")// "BMP","PNG","GIF","JPG"
oBMP:SetBuffer( cBuffer, XBPBMP_FORMAT_PNG ) // _BMP,_PNG,_GIF,_JPG
oBMP:transparentClr := oBMP:getDefaultBgColor()

* PushButton
oButon := XbpPushButton():new()
oButon:caption := oBMP
oButon:create( oDlg, , {10,20}, {70,70} )

* Picture
oBitmap:= XbpStatic():new(oDlg,oDlg, {30,100}, {50,50} )
oBitmap:type := XBPSTATIC_TYPE_BITMAP
oBitmap:caption := oBMP
oBitmap:autoSize := .T.
oBitmap:create()

// Sa oBMP:transparentClr bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna
```

```
* eXpress++
aColor := {0,0,0} // PNG transparent RGB color black
DC_BitmapTransparentColor(aColor)

oBMP := DC_GetBitmap( 1001,,"PNG","RESURS.DLL")// "BMP","PNG","GIF","JPG"

* PushButton
@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3

* Picture
@ 1,1 DCSTATIC TYPE XBPSTATIC_TYPE_BITMAP SIZE 50,50 PIXEL CAPTION oBMP

// Sa DC_BitmapTransparentColor(aColor) bitmapa je transparentna
// na PushButton i na Picture, inače nije transparentna
```



## PROJECT APPLICATION AND RESOURCES

File.**XPJ** (Project file - script)

File.**CH** (Heder file)

File.**ARC** (Resource file - script)

File.**RES** (Resource file)
















File.**PRG** (Source code file)

File.**DLL** (Dynamic-link library file)

File.**EXE** (Executable file - Application)

## Folder: RESOURCE

## Folder: RESOURCE\ICON

Name		
ICON	 USB1.BMP ACDSee Pro 2.0 BMP Image 3,30 KB	 USB1.gif ACDSee Pro 1,69 KB
RESOURCEEXE.XPJ	 USB1.ICO ACDSee Pro 2.0 ICO Image 8,94 KB	 USB1.jpg ACDSee Pro 3,04 KB
RESOURCEDLL.XPJ	 USB1.png ACDSee Pro 2.0 PNG Image 1,59 KB	 USB2.BMP ACDSee Pro 9,05 KB
RESOURCE.CH	 USB2.gif ACDSee Pro 2.0 GIF Image 1,85 KB	 USB2.ICO ACDSee Pro 17,1 KB
RESOURCEEXE.BAT	 USB2.jpg ACDSee Pro 2.0 JPEG Image 2,77 KB	 USB2.png ACDSee Pro 3,85 KB
RESOURCEDLL.BAT	 USB3.BMP ACDSee Pro 2.0 BMP Image 9,05 KB	 USB3.gif ACDSee Pro 1,64 KB
RESOURCE.BAT	 USB3.ICO ACDSee Pro 2.0 ICO Image 17,1 KB	 USB3.jpg ACDSee Pro 3,21 KB
README_RESOURCEEXE.TXT	 USB3.png ACDSee Pro 2.0 PNG Image 2,41 KB	
RESOURCE.RES		
RESOURCE_VIEW2.PRG		
RESOURCE_VIEW1.PRG		
WINDOWSXP.MANIFEST		
RESOURCE.ARC		
RESOURCEDLL.DLL		
RESOURCEEXE.EXE		

**Example: fajl RESURS.ARC**

```
#include "RESURS.CH"

ICON
    ICO_USB1 = "ICON\USB1.ICO"
    ICO_USB2 = "ICON\USB2.ICO"
    ICO_USB3 = "ICON\USB3.ICO"

BITMAP
    BMP_USB1 = "ICON\USB1.BMP"
    BMP_USB2 = "ICON\USB2.BMP"
    BMP_USB3 = "ICON\USB3.BMP"

USERDEF PNG
1001      = FILE "ICON\USB1.PNG"
2001      = FILE "ICON\USB2.PNG"
3001      = FILE "ICON\USB3.PNG"

USERDEF GIF
1002      = FILE "ICON\USB1.GIF"
2002      = FILE "ICON\USB2.GIF"
3002      = FILE "ICON\USB3.GIF"

USERDEF JPG
1003      = FILE "ICON\USB1.JPG"
2003      = FILE "ICON\USB2.JPG"
3003      = FILE "ICON\USB3.JPG"

#define MANIFEST_RESID 1
#define MANIFEST 24
#ifdef __ARC__
    USERDEF MANIFEST
    MANIFEST_RESID = FILE "WINDOWSXP.MANIFEST"
#endif
```

**Example: fajl RESURS.CH**

```
// ICON
#define ICO_USB1 1
#define ICO_USB2 2
#define ICO_USB3 3

// BITMAP
#define BMP_USB1 101
#define BMP_USB2 102
#define BMP_USB3 103

#define MENUITEM_SEPARATOR {NIL,NIL,XBPMENUBAR_MIS_SEPARATOR, 0}
```

**Example: fajl RESURS.BAT ---> RESURS.RES**

```
@ECHO OFF
ARC.EXE RESURS.ARC -v
PAUSE
```

**Example: fajl RESURSEX.XPJ**

```
[PROJECT]
  COMPILE      = xpp
  COMPILE_FLAGS = /q
  DEBUG        = no
  GUI          = yes
  LINKER       = alink
  LINK_FLAGS   =
  RC_COMPILE   = arc
  RC_FLAGS     = /v
  OBJ_DIR      = RESURSEX_____OBJ
  RESURSEX.XPJ

[RESURSEX.XPJ]

  RESURSEX.EXE

[RESURSEX.EXE]

  RESURS.ARC
  RESURS.CH
  RESURS.RES

  DCLIPX.LIB // eXpress++

  RESOURCE_VIEW1.PRG
  RESOURCE_VIEW2.PRG
```

**Example: fajl RESURSEX.BAT ---> RESURSEX.EXE**

```
@ECHO OFF
PBUILD.EXE RESURSEX.XPJ > RESURSEX_____.TXT
Pause
```

**Example: fajl RESURSDLL.XPJ**

```
[PROJECT]
  COMPILE      = xpp
  COMPILE_FLAGS = /q
  DEBUG        = no
  GUI          = yes
  LINKER       = alink
  LINK_FLAGS   =
  RC_COMPILE   = arc
  RC_FLAGS     = /v
  OBJ_DIR      = RESURSDLL_____OBJ
  RESURSDLL.XPJ

[RESURSDLL.XPJ]

  RESURSDLL.DLL

[RESURSDLL.DLL]

COMPILE_FLAGS = -q -dll:dynamic

  RESURS.ARC
  RESURS.CH
  RESURS.RES
```

**Example: fajl RESURSDLL.BAT ----> RESURSDLL.DLL**

```
@ECHO OFF
PBUILD.EXE RESURSDLL.XPJ > RESURSDLL_____.TXT
Pause
```

**Example: fajl RESOURCE\_VIEW1.PRG**

```
/*
* resource_view1.prg
* (29-05-2014)
* © COBA Systems Software
* BAST® Open Source Project
* Business Account Software Technology
* www.cobasystems.com

ICO i BMP resursi se prikazuju na komandnom dugmetu
Xbase++ and eXpres++ code, ver.1.9.355

FUNCTION resource_view()
    STATIC FUNCTION hlp1()
    STATIC FUNCTION hlp2()
    STATIC FUNCTION hlp3()
    STATIC FUNCTION tekst()

FUNCTION gde_exe()
FUNCTION hlp(nn)

*/

#include "Appevent.ch"
#include "Common.ch"
#include "Font.ch"
#include "Gra.ch"
#include "Xbp.ch"
#include "Directory.ch" // za funkciju directory()
#include "xbtsys.ch"

#include "DCDIALOG.CH"
#include "DCGRA.CH"

#include "RESOURCE.CH"

PROCEDURE APPSYS()
    SET CHARSET TO ANSI
    SET DATE GERMAN
    RETURN

PROCEDURE MAIN()
    resource_view1()
    RETURN

FUNCTION gde_exe()
    RETURN strtran(appname(.t.), "\"+appname())

*****
FUNCTION resource_view1()
*****

LOCAL GetList := {}, oDlg, GetOptions, aCUR := {"user32.dll",114}

//----- PREUZIMANJE RESURSA IZ FAJLOVA NA DISKU -----
```

```

@ 0,1 DCSAY "DOWNLOAD RESOURCES FROM THE FILES ON DISK" FONT "10.Arial Bold" SAYSIZE 0

xfolder := gde_exe()+"\ICON"
xbitmap := xfolder + "\" + "USB1.BMP"
xicon := xfolder + "\" + "USB1.ICO"

* oBMP := XbpBitmap():new():create()
* oBMP:LoadFile(xbitmap)
** or:
oBMP := DC_GetBitmap( xbitmap )

@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3 CURSOR aCUR ;
ACTION {|| hlp1() }
//-----
@ 1,10 DCSAY xbitmap SAYSIZE 0
@ 2,10 DCSAY 'oBMP := DC_GetBitmap( "ICON\USB1.BMP" )' SAYSIZE 0 COLOR GRA_CLR_DARKRED
@ 3,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

* oICO := XbpIcon():new():create()
* oICO:LoadFile(xicon)
** or:
oICO := DC_GetBitmap( xicon,,,, .T. )

@ 5,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3 CURSOR aCUR ;
ACTION {|| hlp1() }
//-----
@ 5,10 DCSAY xicon SAYSIZE 0
@ 6,10 DCSAY 'oICO := DC_GetBitmap( "ICON\USB1.ICO", , , , .T.)' SAYSIZE 0 COLOR
GRA_CLR_DARKRED
@ 7,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

//----- PREUZIMANJE RESURSA IZ FAJLOVA NA DISKU -----

//----- PREUZIMANJE RESURSA IZ EXE/RES FAJLA -----

@ 9,1 DCSAY "DOWNLOAD INSTALLED RESOURCE IN EXE FILE" FONT "10.Arial Bold" SAYSIZE 0

* oBMP := BMP_USB2
* oBMP := 102
** or:
oBMP := DC_GetBitmap( 102 )

@ 10,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3 CURSOR aCUR ;
ACTION {|| hlp2() }
//-----
@ 10,10 DCSAY '#include "ICON\RESOURCE.CH"' SAYSIZE 0
@ 11,10 DCSAY 'oBMP := DC_GetBitmap( 102 )' SAYSIZE 0 COLOR GRA_CLR_DARKRED
@ 12,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

* oICO := ICO_USB2
* oICO := 2
** or:
oICO := DC_GetBitmap( 2,,,, .T. )

```

```

    @ 14,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3 CURSOR aCUR ;
        ACTION {|| hlp2() }
//-----
    @ 14,10 DCSAY '#include "ICON\RESOURCE.CH"' SAYSIZE 0
    @ 15,10 DCSAY 'oICO := DC_GetBitmap( 2, , , , .T.)' SAYSIZE 0 COLOR GRA_CLR_DARKRED
    @ 16,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

//----- PREUZIMANJE RESURSA IZ EXE/RES FAJLA -----

//----- PREUZIMANJE RESURSA IZ DLL FAJLA -----

@ 18,1 DCSAY "DOWNLOAD INSTALLED RESOURCE IN DLL FILE" FONT "10.Arial Bold"    SAYSIZE 0

xfolder    := gde_exe()+"\ICON"
xdll       := xfolder + "\" + "RESOURCEDLL.DLL"

    * oBMP := XbpBitmap():new():create()
    * oBMP:Load(xdll,103)
    ** or:
        oBMP := DC_GetBitmap( 103,, "BMP",xdll )

    @ 19,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3 CURSOR aCUR ;
        ACTION {|| hlp3() }
//-----
    @ 19,10 DCSAY xfolder + '\' + 'RESOURCEDLL.DLL' SAYSIZE 0
    @ 20,10 DCSAY 'oBMP := DC_GetBitmap( 103,, "BMP",xdll )' SAYSIZE 0 COLOR GRA_CLR_DARKRED
    @ 21,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

    * oICO := XbpIcon():new():create()
    * oICO:Load(xdll,3)
    ** or:
        oICO := DC_GetBitmap( 3,, "ICO",xdll)

    @ 23,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3 CURSOR aCUR ;
        ACTION {|| hlp3() }
//-----
    @ 23,10 DCSAY xfolder + '\' + 'RESOURCEDLL.DLL' SAYSIZE 0
    @ 24,10 DCSAY 'oICO := DC_GetBitmap( 3,, "ICO",xdll)' SAYSIZE 0 COLOR GRA_CLR_DARKRED
    @ 25,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

//----- PREUZIMANJE RESURSA IZ DLL FAJLA -----

x:=27;y:=1
@ x,y ;
    DCPUSHBUTTON CAPTION "ICO" ;
    SIZE 11,2 ;
    ACTION {|| xbase_ico() }; // resource_view2.prg
    CURSOR aCUR
@ x,y+11 ;
    DCPUSHBUTTON CAPTION "BMP" ;

```



```

        SIZE 11,2 ;
        ACTION {|| xbase_png() }; // resource_view2.prg
        CURSOR aCUR

@ x,y+11+11 ;
        DCPUSHBUTTON CAPTION "PNG-GIF-JPG" ;
        SIZE 13,2 ;
        ACTION {|| resource_view2() }; // resource_view2.prg
        CURSOR aCUR

@ x,y+11+11+13 ;
        DCPUSHBUTTON CAPTION "Help" ;
        ACCELKEY xbeK_F1;
        SIZE 11,2 ;
        ACTION {|| hlp(1) }; // ovde
        CURSOR aCUR

@ x,y+11+11+13+11 ;
        DCPUSHBUTTON CAPTION "Exit" ;
        ACCELKEY xbeK_ESC;
        SIZE 10,2 ;
        ACTION {|| DC_ReadGuiEvent(DCGUI_EXIT_OK,GetList) };
        CURSOR aCUR

DCGETOPTIONS NOMINBUTTON NOMAXBUTTON SAYFONT "8.Consolas Bold"
DCREAD GUI FIT TITLE "RESOURCE FILES: ICO, BMP, ARC, CH, RES, DLL";
        OPTIONS GetOptions ;
        MODAL EVAL {||o|setappwindow(o)}

RETURN nil

*****
STATIC FUNCTION hlp1()
*****
PRIVATE txt1:="", txt2:="", txt3:="", txt4:="",txt5:="" , txt30:=""
PRIVATE txt_1:="", txt_2:="", txt_3:=""
tekst()
tekst := txt_1+txt1+txt30
DC_GuiMemoEdit( tekst,0,0,40,80,.F.,,,,,,"DOWNLOAD RESOURCES FROM THE FILES ON DISK")
RETURN NIL

*****
STATIC FUNCTION hlp2()
*****
PRIVATE txt1:="", txt2:="", txt3:="", txt4:="",txt5:="" , txt31:=""
PRIVATE txt_1:="", txt_2:="", txt_3:=""
tekst()
tekst := txt_2+txt1+txt2+txt31
DC_GuiMemoEdit( tekst,0,0,40,80,.F.,,,,,,"DOWNLOAD INSTALLED RESOURCE IN EXE FILE")
RETURN NIL

*****
STATIC FUNCTION hlp3()
*****
PRIVATE txt1:="", txt2:="", txt3:="", txt4:="",txt5:="" , txt30:=""
PRIVATE txt_1:="", txt_2:="", txt_3:=""
tekst()
tekst := txt_3+txt1+txt2+txt4+txt30
DC_GuiMemoEdit( tekst,0,0,40,80,.F.,,,,,,"DOWNLOAD INSTALLED RESOURCE IN DLL FILE")

```

```

RETURN NIL

*****
STATIC FUNCTION tekst()
*****
LOCAL txt := "", cr := chr(13)+chr(10)

//-----
TEXT INTO txt_1 WRAP cr TRIMMED

PREUZIMANJE RESURSA IZ FAJLOVA NA DISKU:
DOWNLOAD RESOURCES FROM THE FILES ON DISK:
-----

File: C:\RESOURCE\RESOURCE_VIEW.PRG
-----

xfolder := "C:\RESOURCE\ICON"
xbitmap := xfolder + "\" + "USB1.BMP"
xicon := xfolder + "\" + "USB1.ICO"

// Xbase++
oBMP := XbpBitmap():new():create()
oBMP:LoadFile(xbitmap)

// eXpress++
oBMP := DC_GetBitmap( xbitmap )

@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3

// Xbase++
oICO := XbpBitmap():new():create()
oICO:LoadFile(xicon)

// eXpress++
oICO := DC_GetBitmap( xicon,,,, .T. )

@ 5,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3

ENDTEXT
//-----

//-----
TEXT INTO txt_2 WRAP cr TRIMMED

PREUZIMANJE UGRAĐENIH RESURSA U EXE FAJL:
DOWNLOAD INSTALLED RESOURCE IN EXE FILE:
-----

File: C:\RESOURCE\RESOURCE_VIEW.PRG
-----
xexe:="C:\RESOURCE\RESOURCEEXE.EXE"

// Xbase++
oBMP := BMP_USB2
oBMP := 102

// eXpress++
oBMP := DC_GetBitmap( 102 )

@ 10,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3

// Xbase++
oICO := ICO_USB2
oICO := 2

```

```
// eXpress++
oICO := DC_GetBitmap( 2,,,,, .T. )

@ 14,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3

ENDTEXT
//-----

//-----
TEXT INTO txt_3 WRAP cr TRIMMED

PREUZIMANJE RESURSA UGRAĐENIH U DLL FAJL:
DOWNLOAD INSTALLED RESOURCE IN DLL FILE:
-----

File: C:\RESOURCE\RESOURCE_VIEW.PRG
-----
xdll:="C:\RESOURCE\RESOURCEDLL.DLL"

// Xbase++
oBMP := XbpBitmap():new():create()
oBMP:Load(xdll,103)
// or
ndll := DllLoad(xdll)
oBMP := XbpBitmap():new():create()
oBMP:Load(ndll,103)
DllUnload(ndll)

// eXpress++
oBMP := DC_GetBitmap( 103,,"BMP",xdll )

@ 19,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3

// Xbase++
oICO := XbpIcon():new():create()
oICO:Load(xdll,3)

// eXpress++
oICO := DC_GetBitmap( 3,,"ICO",xdll )

@ 23,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3

ENDTEXT
//-----

//-----
TEXT INTO txt1 WRAP cr TRIMMED

-----
C O N T E N T   O F   T H E   P R O J E C T
-----

Folder: C:\RESOURCE\ICON\
-----
ICON:

USB1.ICO
USB2.ICO
USB3.ICO

BITMAP:

USB1.BMP
```

```
USB2.BMP
USB3.BMP

USB1.PNG
USB2.PNG
USB3.PNG

USB1.GIF
USB2.GIF
USB3.GIF

ENDTEXT
//-----

//-----
TEXT INTO txt2 WRAP cr TRIMMED

File: C:\RESOURCE\RESOURCE.ARC
-----
// start
#include "RESOURCE.CH"
ICON
    ICO_USB1 = "ICON\USB1.ICO"
    ICO_USB2 = "ICON\USB2.ICO"
    ICO_USB3 = "ICON\USB3.ICO"
BITMAP
    BMP_USB1 = "ICON\USB1.BMP"
    BMP_USB2 = "ICON\USB2.BMP"
    BMP_USB3 = "ICON\USB3.BMP"

USERDEF PNG
1000      = FILE "ICON\USB1.PNG"
2000      = FILE "ICON\USB2.PNG"
3000      = FILE "ICON\USB3.PNG"

USERDEF GIF
1001      = FILE "ICON\USB1.GIF"
2002      = FILE "ICON\USB2.GIF"
3003      = FILE "ICON\USB3.GIF"
// end

File: C:\RESOURCE\RESOURCE.CH
-----
// start
// ICON
    #define ICO_USB1 1
    #define ICO_USB2 2
    #define ICO_USB3 3
// BITMAP
    #define BMP_USB1 101
    #define BMP_USB2 102
    #define BMP_USB3 103
// end
ENDTEXT
//-----

//-----
TEXT INTO txt30 WRAP cr TRIMMED

File: C:\RESOURCE\RESOURCEEXE.XPJ
-----
// start
[PROJECT]
    COMPILE      = xpp
    COMPILE_FLAGS = /q
    DEBUG        = no
```

```

GUI          = yes
LINKER       = alink
LINK_FLAGS   =
RC_COMPILE   = arc
RC_FLAGS     = /v
OBJ_DIR      = RESOURCEEXE_____OBJ
RESOURCEEXE.XPJ

[RESOURCEEXE.XPJ]
    RESOURCEEXE.EXE
[RESOURCEEXE.EXE]

// FILES: RESOURCES// NO resources
//  RESOURCE.CH    // NO resources
//  RESOURCE.ARC   // NO resources
//  RESOURCE.RES   // NO resources

    DCLIPX.LIB      // eXpress++
    RESOURCE_VIEW1.PRG
    RESOURCE_VIEW2.PRG
// end

File: C:\RESOURCE\RESOURCEEXE.BAT
-----
// start
@ECHO OFF
PBUILD.EXE RESOURCEEXE.XPJ > RESOURCEEXE_____.TXT
pause
// end
ENDTEXT
//-----

//-----
TEXT INTO txt31 WRAP cr TRIMMED

File: C:\RESOURCE\RESOURCEEXE.XPJ
-----
// start
[PROJECT]
    COMPILE        = xpp
    COMPILE_FLAGS   = /q
    DEBUG           = no
    GUI             = yes
    LINKER          = alink
    LINK_FLAGS      =
    RC_COMPILE       = arc
    RC_FLAGS        = /v
    OBJ_DIR         = RESOURCEEXE_____OBJ
    RESOURCEEXE.XPJ

[RESOURCEEXE.XPJ]
    RESOURCEEXE.EXE
[RESOURCEEXE.EXE]

// FILES: RESOURCES
//  RESOURCE.CH
//  RESOURCE.ARC
//  RESOURCE.RES

    DCLIPX.LIB      // eXpress++
    RESOURCE_VIEW1.PRG
    RESOURCE_VIEW2.PRG
// end

File: C:\RESOURCE\RESOURCEEXE.BAT
-----
// start

```

```

@ECHO OFF
PBUILD.EXE RESOURCEEXE.XPJ > RESOURCEEXE_____.TXT
pause
// end
ENDTEXT
//-----

//-----
TEXT INTO txt4 WRAP cr TRIMMED

File: C:\RESOURCE\RESOURCEDLL.XPJ
-----
// start
[PROJECT]
    COMPILE      = xpp
    COMPILE_FLAGS = -q
    DEBUG        = no
    GUI          = yes
    LINKER       = alink
    LINK_FLAGS   =
    OBJ_DIR      = RESOURCEDLL_____OBJ
    RC_COMPILE   = arc
    RC_FLAGS     = -v
    RESOURCEDLL.XPJ

[RESOURCEDLL.XPJ]
    RESOURCEDLL.DLL
[RESOURCEDLL.DLL]
    COMPILE_FLAGS = -q -dll:dynamic
    // FILES: RESOURCES
    RESOURCE.CH
    RESOURCE.ARC
    RESOURCE.RES
// end

File: C:\RESOURCE\RESOURCEDLL.BAT
-----
// start
@echo off
PBUILD.EXE RESOURCEDLL.XPJ > RESOURCEDLL_____.TXT
PAUSE
// end
ENDTEXT
//-----

* c__procitaj(txt,"Uputstvo",,"H","B")
* CLEAR TYPEAHEAD
RETURN NIL

*****
FUNCTION hlp(nn)
*****
LOCAL txt := "", cr := chr(13)+chr(10)
TEXT INTO txt WRAP cr TRIMMED
Complete project
-----

with accompanying source code
and instructions for working

Testing the operation
of Alaska Xbase++ 1.9.355 and

```

```
eXpress++ 1.9
to load resources:
ICO, BMP, PNG, GIF, JPG
and view those pictures
on the command button

ICO,BMP,PNG,GIF,JPG resources
are taken from
ICO,BMP,PNG,GIF,JPG files on the disk,
or EXE file, or DLL file

www.cobasystems.com
COBA Systems Software 2014
Slobodan Stanojevic Cobra

30.05.2014
ENDTEXT
confirmbox(,txt,"Help",XBPMB_OK)
CLEAR TYPEAHEAD
RETURN NIL
```

**Example: fajl RESOURCE\_VIEW2.PRG**

```

/*
* resource_view2.prg
* (29-05-2014)
* © COBA Systems Software
* BAST® Open Source Project
* Business Account Software Technology
* www.cobasystems.com

ICO - resursi se prikazuju na komandnom dugmetu
BMP - resursi se prikazuju na komandnom dugmetu
PNG,GIF,JPG - resursi se prikazuju na komandnom dugmetu

Xbase++ and eXpres++ code, ver.1.9.355

FUNCTION resource_view1()
    STATIC FUNCTION hlp1()
    STATIC FUNCTION hlp2()
    STATIC FUNCTION hlp3()
    STATIC FUNCTION tekst()

PROCEDURE Xbase_png()
    STATIC FUNCTION h1()
    STATIC FUNCTION h2()
    STATIC FUNCTION h3()
    STATIC FUNCTION h0()

PROCEDURE Xbase_ico()
    STATIC FUNCTION xbase_ico_help()

*/

#include "Appevent.ch"
#include "Common.ch"
#include "Font.ch"
#include "Gra.ch"
#include "Xbp.ch"
#include "Directry.ch" // za funkciju directory()

#include "Dll.ch" // za loadresource()
#include "xbtsys.ch"

#include "DCDIALOG.CH"
#include "DCGRA.CH"

#include "RESOURCE.CH"

*****
FUNCTION resource_view2()
*****
LOCAL GetList := {}, oDlg, GetOptions, aCUR := {"user32.dll",114}

//----- PREUZIMANJE RESURSA IZ FAJLOVA NA DISKU -----

* 1001
* 1002
* 1003

```



```

@ 0,1 DCSAY "DOWNLOAD RESOURCES FROM THE FILES ON DISK" FONT "10.Arial Bold" SAYSIZE 0

xfolder    := gde_exe()+"\ICON"

xpng       := xfolder + "\" + "USB1.PNG"
xgif       := xfolder + "\" + "USB1.GIF"
xjpg       := xfolder + "\" + "USB1.JPG"

// PNG

* oBMP := XbpBitmap():new():create()
* oBMP:LoadFile(xpng)

oBMP := DC_GetBitmap( xpng )

@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3 CURSOR aCUR ;
ACTION {|| hlp1() }
//-----
@ 1,10 DCSAY 'xpng='+xpng SAYSIZE 0
@ 2,10 DCSAY 'oBMP := DC_GetBitmap( xpng )' SAYSIZE 0 COLOR GRA_CLR_DARKRED
@ 3,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

// GIF

* oBMP := XbpBitmap():new():create()
* oBMP:LoadFile(xgif)

oBMP := DC_GetBitmap( xgif )

@ 4,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3 CURSOR aCUR ;
ACTION {|| hlp1() }
//-----
@ 4,10 DCSAY 'xgif='+xgif SAYSIZE 0
@ 5,10 DCSAY 'oBMP := DC_GetBitmap( xgif )' SAYSIZE 0 COLOR GRA_CLR_DARKRED
@ 6,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

// JPG

* oBMP := XbpBitmap():new():create()
* oBMP:LoadFile(xjpg)

oBMP := DC_GetBitmap( xjpg )

@ 7,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3 CURSOR aCUR ;
ACTION {|| hlp1() }
//-----
@ 7,10 DCSAY 'xjpg='+xjpg SAYSIZE 0
@ 8,10 DCSAY 'oBMP := DC_GetBitmap( xjpg )' SAYSIZE 0 COLOR GRA_CLR_DARKRED
@ 9,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

//----- PREUZIMANJE RESURSA IZ FAJLOVA NA DISKU -----

//----- PREUZIMANJE RESURSA IZ EXE/RES FAJLA -----

* 2001

```

```

* 2002
* 2003

@ 10,1 DCSAY "DOWNLOAD INSTALLED RESOURCE IN EXE FILE" FONT "10.Arial Bold" SAYSIZE 0

xfolder := gde_exe()
xexe := xfolder + "\" + "RESOURCEEXE.EXE"

// PNG

oBMP := XbpBitmap():new():create()
cBuffer := LoadResource(2001,, "PNG")
oBMP:SetBuffer(cBuffer,XBPBMP_FORMAT_PNG)
oBMP:transparentClr := oBMP:getDefaultBgColor()

* aColor := { 0, 0, 0 } // For PNG transparent color is black
* DC_BitmapTransparentColor( aColor )
* oBMP := DC_GetBitmap( 2001,, "PNG" )
* oBMP := DC_GetBitmap( 2001,, "PNG", xexe)

@ 11,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3 CURSOR aCUR ;
ACTION {|| hlp2() }
//-----
@ 11,10 DCSAY 'USERDEF PNG 2001 = FILE "ICON\USB2.PNG"' SAYSIZE 0
@ 12,10 DCSAY 'oBMP := DC_GetBitmap( 2001,, "PNG" )' SAYSIZE 0 COLOR GRA_CLR_DARKRED
@ 13,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

// GIF

oBMP := XbpBitmap():new():create()
cBuffer := LoadResource(2002,, "GIF")
oBMP:SetBuffer(cBuffer,XBPBMP_FORMAT_GIF)
oBMP:transparentClr := oBMP:getDefaultBgColor()

* aColor := { 0, 0, 0 } // For GIF transparent color is black
* DC_BitmapTransparentColor( aColor )
* oBMP := DC_GetBitmap( 2002,, "GIF" )
* oBMP := DC_GetBitmap( 2002,, "GIF", xexe )

@ 14,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3 CURSOR aCUR ;
ACTION {|| hlp2() }
//-----
@ 14,10 DCSAY 'USERDEF GIF 2002 = FILE "ICON\USB2.GIF"' SAYSIZE 0
@ 15,10 DCSAY 'oBMP := DC_GetBitmap( 2002,, "GIF" )' SAYSIZE 0 COLOR GRA_CLR_DARKRED
@ 16,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

// JPG

oBMP := XbpBitmap():new():create()
cBuffer := LoadResource(2003,, "JPG")
oBMP:SetBuffer(cBuffer,XBPBMP_FORMAT_JPG)

* oBMP := DC_GetBitmap( 2003,, "JPG" ) // no transparency
* oBMP := DC_GetBitmap( 2003,, "JPG", xexe )

@ 17,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3 CURSOR aCUR ;
ACTION {|| hlp2() }
//-----
@ 17,10 DCSAY 'USERDEF JPG 2003 = FILE "ICON\USB2.JPG"' SAYSIZE 0

```

```

@ 18,10 DCSAY 'oBMP := DC_GetBitmap( 2003,, "JPG" )' SAYSIZE 0 COLOR GRA_CLR_DARKRED
@ 19,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

//----- PREUZIMANJE RESURSA IZ EXE/RES FAJLA -----

//----- PREUZIMANJE RESURSA IZ DLL/RES FAJLA -----

* 3001
* 3002
* 3003

@ 20,1 DCSAY "DOWNLOAD INSTALLED RESOURCE IN DLL FILE" FONT "10.Arial Bold" SAYSIZE 0

xfolder := gde_exe()
xdll := xfolder + "\" + "RESOURCEDLL.DLL"

// PNG

* ndll := DllLoad(xdll) // there must be DllUnload()
oBMP := XbpBitmap():new():create()
cBuffer := LoadResource(3001,xdll,"PNG")
oBMP:SetBuffer(cBuffer,XBPBMP_FORMAT_PNG)
oBMP:transparentClr := oBMP:getDefaultBgColor()

* aColor := { 0, 0, 0 } // For PNG transparent color is black
* DC_BitmapTransparentColor( aColor )
* oBMP := DC_GetBitmap( 3001,, "PNG",xdll)

@ 21,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3 CURSOR aCUR ;
ACTION {|| hlp3() }
//-----
@ 21,10 DCSAY 'USERDEF PNG 3001 = FILE "ICON\USB3.PNG"' SAYSIZE 0
@ 22,10 DCSAY 'oBMP := DC_GetBitmap( 3001,, "PNG",ndll )' SAYSIZE 0 COLOR GRA_CLR_DARKRED
@ 23,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

// GIF

* ndll := DllLoad(xdll) // there must be DllUnload()
oBMP := XbpBitmap():new():create()
cBuffer := LoadResource(3002,xdll,"GIF")
oBMP:SetBuffer(cBuffer,XBPBMP_FORMAT_GIF)
oBMP:transparentClr := oBMP:getDefaultBgColor()

* aColor := { 0, 0, 0 } // For PNG transparent color is black
* DC_BitmapTransparentColor( aColor )
* oBMP := DC_GetBitmap( 3002,, "GIF",xdll )

@ 24,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3 CURSOR aCUR ;
ACTION {|| hlp3() }
//-----
@ 24,10 DCSAY 'USERDEF GIF 3002 = FILE "ICON\USB3.GIF"' SAYSIZE 0
@ 25,10 DCSAY 'oBMP := DC_GetBitmap( 3002,, "GIF",ndll )' SAYSIZE 0 COLOR GRA_CLR_DARKRED
@ 26,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

```

```
// JPG

* ndll      := DllLoad(xdll) // there must be DllUnload()
oBMP       := XbpBitmap():new():create()
cBuffer    := LoadResource(3003,xdll,"JPG")
oBMP:SetBuffer(cBuffer,XBPBMP_FORMAT_JPG)
oBMP:transparentClr := oBMP:getDefaultBgColor() // XBPSYSCLR_TRANSPARENT

* oBMP := DC_GetBitmap( 3003,,"JPG",xdll )
* oBMP:transparentClr := oBMP:getDefaultBgColor() // XBPSYSCLR_TRANSPARENT

@ 27,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3 CURSOR aCUR ;
ACTION {|| hlp3() }
//-----
@ 27,10 DCSAY 'USERDEF JPG 3003 = FILE "ICON\USB3.JPG"' SAYSIZE 0
@ 28,10 DCSAY 'oBMP := DC_GetBitmap( 3003,,"JPG",nDll )' SAYSIZE 0 COLOR GRA_CLR_DARKRED
@ 29,10 DCSAY '@ 1,1 DCPUSHBUTTON CAPTION oICO SIZE 8,3' SAYSIZE 0 COLOR GRA_CLR_DARKRED
//-----

//----- PREUZIMANJE RESURSA IZ DLL/RES FAJLA -----

x:=30;y:=1

@ x,y+16+16+2 ;
DCPUSHBUTTON CAPTION "Help" ;
ACCELKEY xbeK_F1;
SIZE 10,2 ;
ACTION {|| hlp(2) };
CURSOR aCUR

@ x,y+16+16+ 2 +10 ;
DCPUSHBUTTON CAPTION "Exit" ;
ACCELKEY xbeK_ESC;
SIZE 10,2 ;
ACTION {|| DC_ReadGuiEvent(DCGUI_EXIT_OK,GetList) };
CURSOR aCUR

DCGETOPTIONS NOMINBUTTON NOMAXBUTTON SAYFONT "8.Consolas Bold"
DCREAD GUI FIT TITLE "RESOURCE FILES: PNG, GIF, JPG, ARC, CH, RES, DLL";
OPTIONS GetOptions ;
MODAL EVAL {||o|setappwindow(o)}

RETURN nil
```

```

*****
STATIC FUNCTION hlp1()
*****
PRIVATE txt1:="", txt2:="", txt3:="", txt4:="",txt5:="" , txt30:=""
PRIVATE txt_1:="", txt_2:="", txt_3:=""
tekst()
tekst := txt_1+txt1+txt30
DC_GuiMemoEdit( tekst,0,0,40,80,.F.,,,,,,"DOWNLOAD RESOURCES FROM THE FILES ON DISK")
RETURN NIL

*****
STATIC FUNCTION hlp2()
*****
PRIVATE txt1:="", txt2:="", txt3:="", txt4:="",txt5:="" , txt31:=""
PRIVATE txt_1:="", txt_2:="", txt_3:=""
tekst()
tekst := txt_2+txt1+txt2+txt31
DC_GuiMemoEdit( tekst,0,0,40,80,.F.,,,,,,"DOWNLOAD INSTALLED RESOURCE IN EXE FILE")
RETURN NIL

*****
STATIC FUNCTION hlp3()
*****
PRIVATE txt1:="", txt2:="", txt3:="", txt4:="",txt5:="" , txt30:=""
PRIVATE txt_1:="", txt_2:="", txt_3:=""
tekst()
tekst := txt_3+txt1+txt2+txt4+txt30
DC_GuiMemoEdit( tekst,0,0,40,80,.F.,,,,,,"DOWNLOAD INSTALLED RESOURCE IN DLL FILE")
RETURN NIL

*****
STATIC FUNCTION tekst()
*****
LOCAL txt := "", cr := chr(13)+chr(10)

//-----
TEXT INTO txt_1 WRAP cr TRIMMED

PREUZIMANJE RESURSA IZ FAJLOVA NA DISKU:
DOWNLOAD RESOURCES FROM THE FILES ON DISK:
-----

File: C:\RESOURCE\RESOURCE_VIEW1.PRG
-----

xfolder  := "C:\RESOURCE\ICON"
xpng     := xfolder + "\" + "USB1.PNG"
xgif     := xfolder + "\" + "USB1.GIF"
xjpg     := xfolder + "\" + "USB1.JPG"

// Xbase++
oBMP := XbpBitmap():new():create()
oBMP:LoadFile(xpng)

// eXpress++
oBMP := DC_GetBitmap( xpng )

@ 1,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3

napomena:

```

```

-----
Kada su PNG i GIF, kao i BMP i JPG
preuzeti iz EXE ili DLL resursa,
ove bitmape nemaju transparentciju.
Za dobijanje transparentije koriste se komande:
note:
-----
When PNG and GIF, and BMP and JPG
download the EXE or DLL resources,
these bitmaps do not have transparency.
To obtain transparency used commands:

// Xbase++
oBMP:transparentClr := oBMP:getDefaultBgColor()
// eXpress++
aColor := { 0, 0, 0 }
DC_BitmapTransparentColor( aColor )

PNG i GIF koji su uzeti iz PNG fajla i GIF fajla
(ovaj primer) zadržavaju svoju transparentiju,
pa se ne koriste gornje komande

PNG and GIF taken from a PNG file and GIF file
(this example) retain their transparency,
so do not use the above commands
ENDTEXT
//-----

//-----
TEXT INTO txt_2 WRAP cr TRIMMED

PREUZIMANJE UGRAĐENIH RESURSA U EXE FAJL:
DOWNLOAD INSTALLED RESOURCE IN EXE FILE:
-----

File: C:\RESOURCE\RESOURCE_VIEW.PRG
-----
xexe:"C:\RESOURCE\RESOURCEEXE.EXE"

// Xbase++

oBMP      := XbpBitmap():new():create()
cBuffer   := LoadResource(2001,"PNG")
oBMP:SetBuffer(cBuffer,XBPBMP_FORMAT_PNG)
oBMP:transparentClr := oBMP:getDefaultBgColor()

// eXpress++

aColor := { 0, 0, 0 } // PNG transparent color is black
DC_BitmapTransparentColor( aColor )

oBMP := DC_GetBitmap( 2001,"PNG" )
// or
oBMP := DC_GetBitmap( 2001,"PNG",xexe)

// or instead DC_BitmapTransparentColor( aColor )
oBMP:transparentClr := oBMP:getDefaultBgColor() // XBPSYSCLR_TRANSPARENT

@ 10,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3

ENDTEXT

```

```
//-----  
  
//-----  
TEXT INTO txt_3 WRAP cr TRIMMED  
  
PREUZIMANJE RESURSA UGRAĐENIH U DLL FAJL:  
DOWNLOAD INSTALLED RESOURCE IN DLL FILE:  
-----  
  
File: C:\RESOURCE\RESOURCE_VIEW.PRG  
-----  
xdll:="C:\RESOURCE\RESOURCEDLL.DLL"  
  
// Xbase++  
  
    ndll      := DllLoad(xdll) // there must be DllUnload()  
    oBMP      := XbpBitmap():new():create()  
    cBuffer   := LoadResource(3001,ndll,"PNG")  
    // or  
    cBuffer   := LoadResource(3001,xdll,"PNG")  
    oBMP:SetBuffer(cBuffer,XBPBMP_FORMAT_PNG)  
    oBMP:transparentClr := oBMP:getDefaultBgColor()  
  
// eXpress++  
  
    aColor := { 0, 0, 0 } // PNG transparent color is black  
    DC_BitmapTransparentColor( aColor )  
  
    oBMP := DC_GetBitmap( 3001,, "PNG",xdll)  
    // or  
    oBMP := DC_GetBitmap( 3001,, "PNG",ndll)  
  
    // or instead DC_BitmapTransparentColor( aColor )  
    oBMP:transparentClr := oBMP:getDefaultBgColor() // XBPSYSCLR_TRANSPARENT  
  
    @ 19,1 DCPUSHBUTTON CAPTION oBMP SIZE 8,3  
  
ENDTEXT  
//-----  
  
//-----  
TEXT INTO txt1 WRAP cr TRIMMED  
  
-----  
C O N T E N T   O F   T H E   P R O J E C T  
-----  
  
Folder: C:\RESOURCE\ICON\  
-----  
  
BITMAP:  
  
USB1.BMP  
USB2.BMP  
USB3.BMP  
  
USB1.PNG  
USB2.PNG  
USB3.PNG  
  
USB1.GIF  
USB2.GIF  
USB3.GIF
```

```

USB1.JPG"
USB2.JPG"
USB3.JPG"

ENDTEXT
//-----

//-----
TEXT INTO txt2 WRAP cr TRIMMED

File: C:\RESOURCE\RESOURCE.ARC
-----
// start
#include "RESOURCE.CH"
ICON
    ICO_USB1 = "ICON\USB1.ICO"
    ICO_USB2 = "ICON\USB2.ICO"
    ICO_USB3 = "ICON\USB3.ICO"
BITMAP
    BMP_USB1 = "ICON\USB1.BMP"
    BMP_USB2 = "ICON\USB2.BMP"
    BMP_USB3 = "ICON\USB3.BMP"

USERDEF PNG
1001      = FILE "ICON\USB1.PNG"
2001      = FILE "ICON\USB2.PNG"
3001      = FILE "ICON\USB3.PNG"

USERDEF GIF
1002      = FILE "ICON\USB1.GIF"
2002      = FILE "ICON\USB2.GIF"
3002      = FILE "ICON\USB3.GIF"

USERDEF JPG
1003      = FILE "ICON\USB1.JPG"
2003      = FILE "ICON\USB2.JPG"
3003      = FILE "ICON\USB3.JPG"

// end

File: C:\RESOURCE\RESOURCE.CH
-----
// start
// ICON
    #define ICO_USB1 1
    #define ICO_USB2 2
    #define ICO_USB3 3
// BITMAP
    #define BMP_USB1 101
    #define BMP_USB2 102
    #define BMP_USB3 103
// end
ENDTEXT
//-----

//-----
TEXT INTO txt30 WRAP cr TRIMMED

File: C:\RESOURCE\RESOURCEEXE.XPJ
-----
// start
[PROJECT]
    COMPILE      = xpp
    COMPILE_FLAGS = /q

```



```

DEBUG          = no
GUI            = yes
LINKER        = alink
LINK_FLAGS    =
RC_COMPILE    = arc
RC_FLAGS      = /v
OBJ_DIR       = RESOURCEEXE_____OBJ
RESOURCEEXE.XPJ

[RESOURCEEXE.XPJ]
    RESOURCEEXE.EXE
[RESOURCEEXE.EXE]

// FILES: RESOURCES      // NO resources
//   ICON\RESOURCE.CH    // NO resources
//   ICON\RESOURCE.ARC   // NO resources
//   ICON\RESOURCE.RES   // NO resources

    DCLIPX.LIB           // eXpress++
    RESOURCE_VIEW.PRG
// end

File: C:\RESOURCE\RESOURCEEXE.BAT
-----
// start
@ECHO OFF
PBUILD.EXE RESOURCEEXE.XPJ > RESOURCEEXE_____.TXT
pause
// end
ENDTEXT
//-----

//-----
TEXT INTO txt31 WRAP cr TRIMMED

File: C:\RESOURCE\RESOURCEEXE.XPJ
-----
// start
[PROJECT]
    COMPILE          = xpp
    COMPILE_FLAGS    = /q
    DEBUG            = no
    GUI              = yes
    LINKER           = alink
    LINK_FLAGS       =
    RC_COMPILE       = arc
    RC_FLAGS         = /v
    OBJ_DIR          = RESOURCEEXE_____OBJ
    RESOURCEEXE.XPJ

[RESOURCEEXE.XPJ]
    RESOURCEEXE.EXE
[RESOURCEEXE.EXE]

// FILES: RESOURCES
//   ICON\RESOURCE.CH
//   ICON\RESOURCE.ARC
//   ICON\RESOURCE.RES

    DCLIPX.LIB           // eXpress++
    RESOURCE_VIEW.PRG
// end

File: C:\RESOURCE\RESOURCEEXE.BAT
-----
// start
@ECHO OFF

```

```

PBUILD.EXE RESOURCEEXE.XPJ > RESOURCEEXE_____.TXT
pause
// end
ENDTEXT
//-----

//-----
TEXT INTO txt4 WRAP cr TRIMMED

File: C:\RESOURCE\RESOURCEDLL.XPJ
-----
// start
[PROJECT]
    COMPILE      = xpp
    COMPILE_FLAGS = -q
    DEBUG        = no
    GUI          = yes
    LINKER       = alink
    LINK_FLAGS   =
    OBJ_DIR      = RESOURCEDLL_____OBJ
    RC_COMPILE   = arc
    RC_FLAGS     = -v
    RESOURCEDLL.XPJ

[RESOURCEDLL.XPJ]
    RESOURCEDLL.DLL
[RESOURCEDLL.DLL]
    COMPILE_FLAGS = -q -dll:dynamic
    // FILES: RESOURCES
    ICON\RESOURCE.CH
    ICON\RESOURCE.ARC
    ICON\RESOURCE.RES
// end

File: C:\RESOURCE\RESOURCEDLL.BAT
-----
// start
@echo off
PBUILD.EXE RESOURCEDLL.XPJ > RESOURCEDLL_____.TXT
PAUSE
// end
ENDTEXT
//-----

* c__procitaj(txt,"Uputstvo",,"H","B")
* CLEAR TYPEAHEAD
RETURN NIL

*****

* PROCEDURE Xbase_png()
*   STATIC FUNCTION h1()
*   STATIC FUNCTION h2()
*   STATIC FUNCTION h3()
*   STATIC FUNCTION h0()

#include "appevent.ch"
#include "common.ch"
*****
PROCEDURE Xbase_png()
*****
LOCAL oDlg, oBut
LOCAL oBitmap

```

```

LOCAL lExit
LOCAL oXbp, nEvent, mp1, mp2

*****
* RESOURCE:
*****
LOCAL xfolder := gde_exe()+"\ICON"
LOCAL xpng := xfolder + "\" + "USB1.PNG"
LOCAL xgif := xfolder + "\" + "USB1.GIF"
LOCAL xjpg := xfolder + "\" + "USB1.JPG"
*
* RESOURCE.ARC
*
*ICON
*      ICO_USB1 = "ICON\USB1.ICO"
*      ICO_USB2 = "ICON\USB2.ICO"
*      ICO_USB3 = "ICON\USB3.ICO"
*
*BITMAP
*      BMP_USB1 = "ICON\USB1.BMP" #define BMP_USB1 101
*      BMP_USB2 = "ICON\USB2.BMP" #define BMP_USB1 102
*      BMP_USB3 = "ICON\USB3.BMP" #define BMP_USB1 103
*
* USERDEF PNG
* 1001      = FILE "ICON\USB1.PNG"
* 1002      = FILE "ICON\USB2.GIF"
* 1003      = FILE "ICON\USB3.JPG"
* USERDEF GIF
* 1002      = FILE "ICON\USB1.GIF"
* 2002      = FILE "ICON\USB2.GIF"
* 3002      = FILE "ICON\USB3.GIF"
*
* USERDEF JPG
* 1003      = FILE "ICON\USB1.JPG"
* 2003      = FILE "ICON\USB2.JPG"
* 3003      = FILE "ICON\USB3.JPG"
*****
* RESOURCE:
*****

oBitmap := ExXbpBitmap():new():create() // korisnička funkcija
oBitmap:Load("RESOURCEDLL.DLL",3003,"JPG") // omogućava ovu sintaksu
oBitmap:transparentClr := oBitmap:getDefaultBgColor()

oDlg := XbpDialog():new( , , {0, 0}, {400, 300}, , .T. )
oDlg:title := "CLASS ExXbpBitmap (BMP,PNG,GIF,JPG)"
oDlg:taskList := .T.
oDlg:close := { || lExit := .T. }
oDlg:DrawingArea:bitmap := oBitmap
oDlg:create()

//-----
oIcon := XbpIcon():new():create()
oIcon:Load("RESOURCEDLL.DLL",1)

oBut := XbpPushButton():new()
oBut:caption := oIcon
oBut:create( oDlg, , {10,20}, {70,70} )
oBut:activate:= {|| h1() }
//-----
oBitmap := ExXbpBitmap():new():create() // korisnička funkcija
oBitmap:Load("RESOURCEDLL.DLL",101) // omogućava ovu sintaksu
oBitmap:transparentClr := oBitmap:getDefaultBgColor()

```

```

        oBut := XbpPushButton():new()
        oBut:caption := oBitmap
        oBut:create( oDlg, , {10+74,20}, {70,70} )
        oBut:activate:= {|| h2() }
//-----
        oBitmap := ExXbpBitmap():new():create()      // korisnička funkcija
        oBitmap:Load("RESOURCE.DLL",1001,"PNG") // omogućava ovu sintaksu... - bez
LoadResource() i SetBuffer()
        oBitmap:transparentClr := oBitmap:getDefaultBgColor()

        oBut := XbpPushButton():new()
        oBut:caption := oBitmap
        oBut:create( oDlg, , {10+74+74,20}, {70,70} )
        oBut:activate:= {|| h3() }
//-----
        oBitmap := ExXbpBitmap():new():create()      // korisnička funkcija
        oBitmap:Load("RESOURCE.DLL",2002,"GIF") // omogućava ovu sintaksu... - bez
LoadResource() i SetBuffer()
        oBitmap:transparentClr := oBitmap:getDefaultBgColor()

        oBut := XbpPushButton():new()
        oBut:caption := oBitmap
        oBut:create( oDlg, , {10+74+74+74,20}, {70,70} )
        oBut:activate:= {|| h0() }
//-----

        oDlg:show()
        SetAppFocus( oDlg )

        nEvent:= xbe_None
        lExit := .F.
        DO WHILE !lExit
            nEvent := AppEvent( @mp1, @mp2, @oXbp )
            oXbp:handleEvent( nEvent, mp1, mp2 )
        ENDDO
        oDlg:destroy()

RETURN

*****
CLASS ExXbpBitmap FROM XbpBitmap
    EXPORTED:
        METHOD Load
ENDCLASS

METHOD ExXbpBitmap:Load(cDll,nID,cType)
LOCAL xResource
LOCAL lSuccess

    DEFAULT cDll TO "RESOURCE.DLL"

    IF EMPTY(cType)

        * this is the standard :load
        * which supports BMP

        lSuccess :=:XbpBitmap:load(cDll,nID)

    ELSE

        * when it comes to GIF, JPG or PNG
        * we can load it from a resource file
        * per LoadResource() function call

        xResource := LoadResource(nID,cDll,cType)
        lSuccess := !EMPTY(xResource)

```

```

::setBuffer(xResource)

ENDIF
RETURN (lSuccess)
* Copyright autor: greetings by OHR, Jimmy
*****

*****
STATIC FUNCTION h1()
*****
LOCAL txt := "", cr := chr(13)+chr(10)
TEXT INTO txt WRAP cr TRIMMED
// BACKGROUND (JPG):

oBitmap := ExXbpBitmap():new():create() // UserFunc
oBitmap:Load("RESOURCEDLL.DLL",3003,"JPG") // USERDEF (resource.arc)
oBitmap:transparentClr := oBitmap:getDefaultBgColor()

oDlg := XbpDialog():new( , , {0, 0}, {400, 300}, , .T. )
oDlg:title := "ExXbpBitmap() USERDEF bitmap test"
oDlg:taskList := .T.
oDlg:close := { || lExit := .T. }
oDlg:DrawingArea:bitmap := oBitmap
oDlg:create()

// PUSHBUTTON (ICO)

oIcon := XbpIcon():new():create()
oIcon:Load("RESOURCEDLL.DLL",1) // ICON (resource.arc)

oBut := XbpPushButton():new()
oBut:caption := oIcon
oBut:create( oDlg, , {10,20}, {100,100} )
oBut:activate:= {|| h1() }

ENDTEXT

DC_GuiMemoEdit( txt,0,0,40,80,.F.,,,,,,"ICON (ICO)")
// confirmbox(,txt,"ICON ICO",XBPMB_OK)
CLEAR TYPEAHEAD
RETURN NIL

*****
STATIC FUNCTION h2()
*****
LOCAL txt := "", cr := chr(13)+chr(10)
TEXT INTO txt WRAP cr TRIMMED
// BACKGROUND (JPG):

oBitmap := ExXbpBitmap():new():create() // UserFunc
oBitmap:Load("RESOURCEDLL.DLL",3003,"JPG") // USERDEF (resource.arc)
oBitmap:transparentClr := oBitmap:getDefaultBgColor()

oDlg := XbpDialog():new( , , {0, 0}, {400, 300}, , .T. )
oDlg:title := "ExXbpBitmap() USERDEF bitmap test"
oDlg:taskList := .T.
oDlg:close := { || lExit := .T. }
oDlg:DrawingArea:bitmap := oBitmap
oDlg:create()

// PUSHBUTTON (BMP)

```

```

oBitmap := ExXbpBitmap():new():create() // UserFunc
oBitmap:Load("RESOURCEDLL.DLL",101) // BITMAP (resource.arc)
oBitmap:transparentClr := oBitmap:getDefaultBgColor()

oBut := XbpPushButton():new()
oBut:caption := oBitmap
oBut:create( oDlg, , {10+104,20}, {100,100} )
oBut:activate:= {|| h2() }

ENDTEXT
DC_GuiMemoEdit( txt,0,0,40,80,.F.,,,,,,"BITMAP (BMP)")
//confirmbox(txt,"BITMAP BMP",XBPMB_OK)
CLEAR TYPEAHEAD
RETURN NIL

*****
STATIC FUNCTION h3()
*****
LOCAL txt := "", cr := chr(13)+chr(10)
TEXT INTO txt WRAP cr TRIMMED
// BACKGROUND (JPG):

oBitmap := ExXbpBitmap():new():create() // UserFunc
oBitmap:Load("RESOURCEDLL.DLL",3003,"JPG") // USERDEF (resource.arc)
oBitmap:transparentClr := oBitmap:getDefaultBgColor()

oDlg := XbpDialog():new( , , {0, 0}, {400, 300}, , .T. )
oDlg:title := "ExXbpBitmap() USERDEF bitmap test"
oDlg:taskList := .T.
oDlg:close := { || lExit := .T. }
oDlg:DrawingArea:bitmap := oBitmap
oDlg:create()

// PUSHBUTTON (PNG,GIF,JPG)

oBitmap := ExXbpBitmap():new():create() // UserFunc
oBitmap:Load("RESOURCEDLL.DLL",1001,"PNG") // USERDEF (resource.arc)
oBitmap:transparentClr := oBitmap:getDefaultBgColor()

oBut := XbpPushButton():new()
oBut:caption := oBitmap
oBut:create( oDlg, , {10+104+104,20}, {100,100} )
oBut:activate:= {|| h3() }

ENDTEXT
DC_GuiMemoEdit( txt,0,0,40,80,.F.,,,,,,"USERDEF (PNG,GIF,JPG)")
//confirmbox(txt,"USERDEF PNG",XBPMB_OK)
CLEAR TYPEAHEAD
RETURN NIL

*****
STATIC FUNCTION h0()
*****
LOCAL txt := "", cr := chr(13)+chr(10)
TEXT INTO txt WRAP cr TRIMMED
CLASS ExXbpBitmap FROM XbpBitmap
EXPORTED:
METHOD Load
ENDCLASS

METHOD ExXbpBitmap:Load(cDll,nID,cType)
LOCAL xResource
LOCAL lSuccess

DEFAULT cDll TO "RESOURCEDLL.DLL"

```

```

IF EMPTY(cType)

    * this is the standard :load
    * which supports BMP

    lSuccess := ::XbpBitmap:load(cDll,nID)

ELSE

    * when it comes to GIF, JPG or PNG
    * we can load it from a resource file
    * per LoadResource() function call

    xResource := LoadResource(nID,cDll,cType)
    lSuccess := !EMPTY(xResource)
    ::setBuffer(xResource)

ENDIF

RETURN (lSuccess)

* Copyright autor: greetings by OHR, Jimmy
ENDTEXT
DC_GuiMemoEdit( txt,0,0,40,80,.F.,,,,,,"CLASS ExXbpBitmap (BMP,PNG,GIF,JPG)")
//confirmbox(txt,"CLASS ExXbpBitmap",XBPMB_OK)
CLEAR TYPEAHEAD
RETURN NIL

* PROCEDURE Xbase_ico()
* STATIC FUNCTION xbase_ico_help()

#include "appevent.ch"
#include "common.ch"
#include "RESOURCE.CH"
*****
PROCEDURE Xbase_ico()
*****
LOCAL oDlg, oBut
LOCAL oBitmap
LOCAL lExit
LOCAL oXbp, nEvent, mp1, mp2

oDlg := XbpDialog():new( , , {0, 0}, {300, 200}, , .T. )
oDlg:title := "RESOURCE ICON (ICO)"
oDlg:taskList := .T.
oDlg:close := { || lExit := .T. }
oDlg:create()

*-----
* XbpStatic
*-----

* XbpStatic icon from exe

oIcon1 := XbpStatic():new(oDlg,oDlg,{30,100}, {32,32} )
oIcon1:type := XBPSTATIC_TYPE_ICON
oIcon1:caption := 1 // numerik - resource index
oIcon1:autoSize := .T.
oIcon1:create()

* XbpStatic icon from DLL

```

```

oIcon2      := XbpStatic():new(oDlg,oDlg,{100,100}, {32,32} )
oIcon2:type  := XBPSTATIC_TYPE_ICON
oIcon2:SetCaption( 2, "RESOURCEDLL.DLL") // numerik - resource index
oIcon2:autoSize := .T.
oIcon2:create()

oIcon3      := XbpStatic():new(oDlg,oDlg,{170,100}, {32,32} )
oIcon3:type  := XBPSTATIC_TYPE_ICON
oIcon3:SetCaption( 3, "RESOURCEDLL.DLL") // numerik - resource index
oIcon3:autoSize := .T.
oIcon3:create()

*-----
* XbpPushButton
*-----

* XbpPushButton icon from exe

    oICO := XbpIcon():new():create()
    oICO:Load(nil,1) // must be object

    oBut := XbpPushButton():new()
    oBut:caption := oICO
    oBut:create( oDlg, , {10,20}, {70,70} )
    oBut:activate:= {|| nil }

* XbpPushButton icon from dll

    oICO := XbpIcon():new():create()
    oICO:Load("RESOURCEDLL.DLL",2) // must be object

    oBut := XbpPushButton():new()
    oBut:caption := oICO
    oBut:create( oDlg, , {10+74,20}, {70,70} )
    oBut:activate:= {|| nil }

* -----

    oBut := XbpPushButton():new()
    oBut:caption := "Help"
    oBut:create( oDlg, , {10+74+74,20}, {70,70} )
    oBut:activate:= {|| xbase_ico_help() }

* -----

oDlg:show()
SetAppFocus( oDlg )

nEvent:= xbe_None
lExit := .F.
DO WHILE !lExit
    nEvent := AppEvent( @mp1, @mp2, @oXbp )
    oXbp:handleEvent( nEvent, mp1, mp2 )
ENDDO
oDlg:destroy()

RETURN

*****
STATIC FUNCTION xbase_ico_help()
*****
LOCAL txt := "", cr := chr(13)+chr(10)
TEXT INTO txt WRAP cr TRIMMED

#include "appevent.ch"
#include "common.ch"

```



```

#include "RESOURCE.CH"
*****
PROCEDURE Xbase_ico()
*****
LOCAL oDlg, oBut
LOCAL oBitmap
LOCAL lExit
LOCAL oXbp, nEvent, mp1, mp2

    oDlg := XbpDialog():new( , , {0, 0}, {300, 200}, , .T. )
    oDlg:title := "RESOURCE ICON (ICO)"
    oDlg:taskList := .T.
    oDlg:close := { || lExit := .T. }
    oDlg:create()

*-----
*   XbpStatic
*-----

*   XbpStatic icon from exe

oIcon1      := XbpStatic():new(oDlg,oDlg,{30,100}, {32,32} )
oIcon1:type  := XBPSTATIC_TYPE_ICON
oIcon1:caption := 1    // numerik - resource index
oIcon1:autoSize := .T.
oIcon1:create()

*   XbpStatic icon from DLL

oIcon2      := XbpStatic():new(oDlg,oDlg,{100,100}, {32,32} )
oIcon2:type  := XBPSTATIC_TYPE_ICON
oIcon2:SetCaption( 2, "RESOURCEDLL.DLL") // numerik - resource index
oIcon2:autoSize := .T.
oIcon2:create()

*-----
*   XbpPushButton
*-----

*   XbpPushButton icon from exe

oICO := XbpIcon():new():create()
oICO:Load(nil,1) // must be object

oBut := XbpPushButton():new()
oBut:caption := oICO
oBut:create( oDlg, , {10,20}, {70,70} )
oBut:activate:= {|| nil }

*   XbpPushButton icon from dll

oICO := XbpIcon():new():create()
oICO:Load("RESOURCEDLL.DLL",2) // must be object

oBut := XbpPushButton():new()
oBut:caption := oICO
oBut:create( oDlg, , {10+74,20}, {70,70} )
oBut:activate:= {|| nil }

*   -----

oBut := XbpPushButton():new()
oBut:caption := "Help"
oBut:create( oDlg, , {10+74+74,20}, {70,70} )
oBut:activate:= {|| xbase_ico_help() }

*   -----

```

```
oDlg:show()
SetAppFocus( oDlg )

nEvent:= xbe_None
lExit := .F.
DO WHILE !lExit
    nEvent := AppEvent( @mp1, @mp2, @oXbp )
    oXbp:handleEvent( nEvent, mp1, mp2 )
ENDDO
oDlg:destroy()


RETURN
ENDTEXT
DC_GuiMemoEdit( txt,0,0,40,80,.F.,,,,,,,,,,"RESOURCE ICON (ICO)")
RETURN NIL
```

**Fajl: WINDOWSXP.MANIFEST**

```
<?xml version="1.0" encoding="UTF-8" standalone="yes"?>
<assembly xmlns="urn:schemas-microsoft-com:asm.v1" manifestVersion="1.0">
  <assemblyIdentity
    name="COBA Systems Software.CSYSTEMS 2014"
    processorArchitecture="x86"
    version="10.0.0.0"
    type="win32"/>
  <description>Programi za knjigovodstvo</description>
  <dependency>
    <dependentAssembly>
      <assemblyIdentity
        type="win32"
        name="Microsoft.Windows.Common-Controls"
        version="6.0.0.0"
        processorArchitecture="x86"
        publicKeyToken="6595b64144ccf1df"
        language="*"
      />
    </dependentAssembly>
  </dependency>
</assembly>
```

## IZGLED RESOURCE.EXE APLIKACIJE


Demonstracija upotrebe resursa ICO i BMP fajlova



## Demonstracija upotrebe ICO fajlova - ikona




## Demonstracija uprošćenje upotrebe BMP, PNG, GIF, JPG resursa putem korisničke funkcije (klase) ExXbpBitmap()



## Demonstracija upotrebe PNG, GIF, JPG resursa



Click miša na bilo koje komandno dugme daje uputstvo za dobijanje resursa na tom dugmetu



Da bi se startovala prikazana aplikacija potrebni su vam Run-Time DLL fajlovi za Alaska Xbase++ 1.9.355 i za eXpress++ 1.9 build 259. Možete ih preuzeti (kao demo verzije) sa sajtova proizvođača:

[www.alaska-software.com](http://www.alaska-software.com)

[www.donnay-software.com](http://www.donnay-software.com)

Za sva pitanja, za konsultacije i za ostale tekstove pogledajte materijal na sajtu

[www.cobasystems.com](http://www.cobasystems.com)

i javite se na

[cobasystems@gmail.com](mailto:cobasystems@gmail.com)

[cobasystemsoftware@gmail.com](mailto:cobasystemsoftware@gmail.com)